
Аннотация.  В данной работе предлагается метод моделирования раскрашенных сетей Пет-
ри на основе онтологий и языка SPARQL. Рассматривается онтология раскрашенных сетей 
Петри, SPARQL-запросы, моделирующие динамику работы, а также вопросы построения 
приложений для интерпретации сетевых моделей. 

Abstract.  This paper proposes a method for colored Petri nets modelling based on ontologies and 
the SPARQL language. The ontology of colored of Petri nets, SPARQL queries for modelling the 
dynamics of functioning, as well as issues of building applications to simulate network models are 
considered. 

Раскрашенные сети Петри (РСП) широко используются  как средство моделирования 

систем, в которых понятия параллельности и событийного взаимодействия являются ключе-
выми [1]. Данный класс сетей Петри отличает наличие свойств, присущих высокоуровневым 
языкам программирования, при сравнительно компактной форме параметризуемых моделей. 
Это, в частности, делает их подходящими для использования при моделировании, проекти-
ровании и реализации распределенных вычислительных, управляющих и коммуникацион-
ных систем в промышленном масштабе. Для представления знаний в рамках определенной 
предметной области, допускающего применение логики, а также позволяющего использо-
вать эти знания повторно, используются онтологии [2]. Они являются основой развивающей-
ся концепции семантического web, которая расширяет возможности машинной интерпрета-
ции данных в сети. Онтологическое представление позволяет определить модель РСП в дек-
ларативном формате, что упрощает процесс интеграции с другими моделями и приложения-
ми семантической сети. Одним из стандартов семантического web, рекомендуемых консор-
циумом W3C, является язык запросов SPARQL [3], использующийся для работы с данными в 
формате RDF. Он предоставляет широкие возможности по извлечению онтологических опи-
саний в форме триплетов и манипуляции ими, что делает его пригодным для моделирования 
динамических систем.  

В данной работе рассматривается моделирование РСП с использованием онтологий и 
SPARQL-запросов. Используемая онтология РСП на уровне концепций приведена на рисун-
ке 1,а. Простой пример РСП представлен на рисунке 1,б. Онтология РСП включает в себя: 
классы Place (Позиция), Arc (Дуга), Transition (Переход), Marking (Маркировка позиции), 
Token (Метка), Attribute (Атрибут), Pattern (Шаблон дуги), Variable (Переменная шаблона) и 
Condition (Сторожевое условие перехода); объектные свойства: comes_from и comes_to (для 
входного и выходного узла дуги), has_marking (для маркировки позиции), has_token (для ме-
ток), has_attribute (для атрибутов меток), has_pattern (для шаблона), has_variable (для пере-
менных шаблона), has_condition (для условия перехода); свойства по данным: has_data (для 
данных маркера и условия перехода), has_index (для определения позиции атрибу-
та/переменной в кортеже) и has_name (для задания имени переменной). Атрибуты меток и 

переменные шаблонов упорядочены в кортеж. Данная онтология была сформирована в ре-
дакторе онтологий Protégé [4]. 

 

 
 

а) 
 

 
 

б) 
 
 

Рис. 1. Представления раскрашенных сетей Петри: а – онтология на уровне TBox;  
б – графическое представление (пример) 

 
Сам процесс моделирования представляет собой симуляцию срабатываний переходов 

посредством выполнения цепочек SPARQL-запросов (транзакций) к онтологии. Фрагменты 
этих запросов приводятся ниже. 

Первый запрос позволяет получить данные по переходам РСП. 
 

PREFIX : <http://www.semanticweb.org/baker/ontologies/2020/9/OntoNet-CPN-onlotogy#> 
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> 
SELECT ?transition ?condition_data ?variables 
WHERE { 
  ?transition rdf:type :Transition. 
  ?transition :has_condition ?condition. 
  ?condition :has_data ?condition_data 
  { 
    SELECT ?transition (GROUP_CONCAT(?var_i_name; SEPARATOR=",") as ?variables) 
    WHERE {{ 
        SELECT DISTINCT ?transition ?var_i_name 
        WHERE { 
          ?arc_i :comes_to ?transition; 
                     :has_pattern ?patt_i. 
          ?patt_i :has_variable ?var_i. 
          ?var_i :has_name ?var_i_name. 
        } 
    }} 
    GROUP BY ?transition 
  } 
} 

В результате выполнения данного запроса получаем URI переходов, их сторожевые 
условия на языке SPARQL и список переменных, использующихся в шаблонах входных дуг 
каждого перехода. 

С информацией по переходам становится возможным определение факта разрешенно-
сти перехода при определенных комбинациях меток во входных позициях. Общий вид 
SPARQL-запроса, отвечающего за это, приведен ниже. 

 

PREFIX : <http://www.semanticweb.org/baker/ontologies/2020/9/OntoNet-CPN-onlotogy#> 
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> 
SELECT ?place_i ?token_i ?values <значения переменных...> 
WHERE { 
  <выборка значений переменных...> 
  BIND((<сторожевое условие перехода>) as ?res) 
  ?arc_i :comes_to <URI перехода>; 
             :comes_from ?place_i; 
             :has_pattern ?patt_i. 
  ?place_i :has_marking ?marking_i. 
  ?marking_i :has_token ?token_i.  
  <выделение значений переменных, сгруппированных по меткам> 
  <проверка корректного сопоставления значения переменной для всех входных дуг...> 
  BIND((<конкатенация значений переменных в строку через запятую>) as ?values) 
} 
GROUP BY ?place_i ?token_i ?res ?values <значения переменных...> 
HAVING (?res = true) 
ORDER BY ?values 

 

Данный запрос определяет комбинации меток, при которых разрешен переход, а так-
же осуществляет согласование переменных шаблонов входных дуг и значений атрибутов ме-
ток из входных позиций. Здесь и далее в угловые скобки заключены части запроса, форми-
руемые динамически на основе данных из предыдущего запроса: идентификатора перехода, 
сторожевого условия перехода и списка переменных. 

Рассмотрим составляющие этого запроса в отдельности на примере простой РСП, 
приведенной выше. Поиск значений осуществляем для перехода t1. Выборка значения и ин-
декса переменной x для нее будет выглядеть следующим образом. 

 

{SELECT DISTINCT ?x ?x_index 
  WHERE { 
    ?arc_i :comes_to :t1; 
               :comes_from ?place_i; 
               :has_pattern ?patt_i. 
    ?place_i :has_marking ?marking_i. 
    ?marking_i :has_token ?token_i.  
    ?token_i :has_attribute ?attr_i. 
    ?attr_i :has_data ?x; 
               :has_index ?x_index. 
    ?patt_i :has_variable ?var_i. 
    ?var_i :has_name ?var_i_name; 
               :has_index ?var_i_index. 
    FILTER(?var_i_name = "x" && ?x_index = ?var_i_index) 
}} 

 

В результате получаем все возможные значения переменной x, а также их индексы в 
кортежах атрибутов в метках входных позиций. Выделение известных значений и индексов 
переменных x и y, сгруппированных по меткам, представлено ниже. 

 

{ 
  SELECT ?token_i ?patt_i  
                  (COUNT(?attr_i) as ?attr_i_count) (COUNT(?var_i) as ?var_i_count)  
                  ?x ?x_index ?y ?y_index 

  WHERE { 
    ?patt_i :has_variable ?var_i. 
    ?token_i :has_attribute ?attr_i. 
    OPTIONAL { 
      SELECT ?token_i ?x ?x_index 
      WHERE { 
        ?token_i :has_attribute ?attr_i. 
        ?attr_i :has_data ?x; 
                   :has_index ?x_index. 
      } 
    } 
    FILTER(NOT EXISTS { 
      ?token_i :has_attribute ?attrx_i. 
      ?attrx_i :has_data ?_var; 
                   :has_index ?_var_index. 
      FILTER ((?_var != ?x || ?_var_index != ?x_index) && (?_var != ?y || ?_var_index != ?y_index)) 
    }) 
  } 
  GROUP BY ?token_i ?patt_i ?x ?x_index ?y ?y_index 
  HAVING (?attr_i_count = ?var_i_count) 
} 

 

Данный фрагмент запроса ограничивает комбинации значений переменных по мет-
кам, используя ранее извлеченные данные об их индексах в пределак кортежей атрибутов. 
Проверка корректного сопоставления значения переменной x для всех входных дуг приведе-
на далее. 

 

FILTER (!BOUND(?x) || NOT EXISTS { 
  SELECT ?place2_i ?x (COUNT(?attr2_i_data) as ?x2_count) 
  WHERE { 
    ?arc2_i :comes_to :t1; 
            :comes_from ?place2_i; 
            :has_pattern ?patt2_i. 
    ?place2_i :has_marking ?marking2_i. 
    ?marking2_i :has_token ?token2_i. 
    ?token2_i :has_attribute ?attr2_i. 
    ?attr2_i :has_index ?attr2_i_index. 
    ?patt2_i :has_variable ?var2_i. 
    ?var2_i :has_name ?var2_i_name; 
            :has_index ?var2_i_index. 
    FILTER(?var2_i_name = "x" && ?attr2_i_index = ?var2_i_index) 
    OPTIONAL { 
      ?attr2_i :has_data ?attr2_i_data; 
               FILTER(?attr2_i_data = ?x) 
    } 
  } 
  GROUP BY ?place2_i ?x 
  HAVING (?x2_count = 0) 
}) 

 

В данном случае проверяется факт наличия аналогичного значения переменной x в 
других входных позициях, где шаблон дуги содержит эту переменную. 

Переход не разрешен, если для него не существует ни одного варианта сопоставления 
значений переменным шаблонов входных дуг. Общий вид SPARQL-запроса, выполняющего 
разрешенный переход, представлен ниже. 

 

DELETE { 
  ?marking_i :has_token ?token_i. 
  ?token_i rdf:type :Token; 

           :has_attribute ?token_i_attr. 
  ?token_i_attr rdf:type :Attribute; 
                :has_data ?token_i_attr_data; 
                :has_index ?token_i_attr_index. 
} 
INSERT { 
  ?marking_o :has_token ?token_new. 
  <формирование меток в выходных позициях> 
} 
WHERE { 
  { 
    VALUES ?token_i { <URI меток, проходящих через переход...> } # <-- :token1-1 :token2-1 
    ?arc_i :comes_to <URI перехода>; 
               :comes_from ?place_i; 
               :has_pattern ?patt_i. 
    ?place_i :has_marking ?marking_i. 
    ?marking_i :has_token ?token_i. 
    ?token_i :has_attribute ?token_i_attr. 
    ?token_i_attr :has_data ?token_i_attr_data; 
                          :has_index ?token_i_attr_index. 
  } UNION { 
    ?arc_o :comes_from <URI перехода>; 
                :comes_to ?place_o; 
                :has_pattern ?patt_o. 
    ?place_o :has_marking ?marking_o. 
    BIND(UUID() as ?token_new) 
    <выборка атрибутов по шаблонам выходных дуг...> 
  } 
} 

 

Данный запрос удаляет метки, определенные в ходе выполнения предыдущего запро-
са, из выходных позиций, и формирует метки в выходных позициях на основе шаблонов вы-
ходных дуг. Идентификаторы новых меток формируются с помощью функции UUID. 

Аналогично, рассмотрим составляющие этого запроса. Фрагмент запроса для форми-
рования данных по новым меткам выходных позиций при наличии сопоставленных пере-
менным x и y значений (“a” для x и “b” для y) приведен ниже: 

 

?token_new rdf:type :Token; 
                    :has_attribute ?x; 
                    :has_attribute ?y. 
?x rdf:type :Attribute; 
                   :has_data "a"; 
                   :has_index ?x_index. 
?y rdf:type :Attribute; 
                  :has_data "b"; 
                  :has_index ?y_index. 

 

Здесь в онтологию вставляются (insert) данные о новых метках и их атрибутах. Вы-
борка атрибутов для переменной x производится следующим образом. 

 

OPTIONAL { 
  SELECT ?patt_o ?x ?x_index 
  WHERE { 
    ?patt_o :has_variable ?var_o. 
    ?var_o :has_name ?var_o_name; 
           :has_index ?x_index. 
    FILTER(?var_o_name = "x") 
    BIND(UUID() as ?x) 
  } 
} 

В данном фрагменте, исходя из шаблонов выходных дуг, выбираются сформирован-
ные идентификаторы атрибутов, а также индексы для будущих кортежей в составе меток. 

Для  получения маркировки РСП на текущем шаге моделирования используется сле-
дующий SPARQL-запрос. 

 

SELECT ?place ?token (GROUP_CONCAT(?data;SEPARATOR=",") AS ?token_data) 
WHERE { 
  ?place rdf:type :Place. 
  ?place :has_marking ?marking. 
  ?marking :has_token ?token. 
  OPTIONAL { 
    ?token :has_attribute ?attr. 
    {SELECT ?attr ?data 
      WHERE { 
        ?attr :has_data ?data. 
        ?attr :has_index ?index. 
      } ORDER BY ?index 
  }} 
} 
GROUP BY ?place ?token 
ORDER BY ?place 

 

Он позволяет получить позиции РСП и метки, содержащиеся в них. 
На практике данный подход был применен при разработке клиент-серверной системы 

интерпретации РСП под названием OntoNet. Она позволяет пользователю загружать на сер-
вер онтологию РСП и моделировать ее выполнение в ручном режиме. В качестве сервера ис-
пользуется Apache Jena Fuseki Server [5], взаимодействие организовано по протоколу HTTP. 
Клиент был разработан на языках JavaScript/TypeScript. Он спроектирован с использованием 
модульной архитектуры, поэтому отдельные его модули могут использоваться в других про-
ектах. Каждый модуль имеет свой интерфейс взаимодействия, предоставляющий широкий 
диапазон методов для получения состояния сети, определения разрешенных переходов, по-
шагового моделирования и т.д. Состояние РСП выводится в виде таблицы, отражающей 
маркировку позиций, однако представление отделено от логики системы и может быть заме-
нено на более продвинутое или специализированное. 

Направлением дальнейших исследований является введение модульности и иерархии 
в «онтологические» РСП, развитие инструментальной системы OntoNet для возможности 
реализации на ее основе распределенных вычислительных и управляющих систем в контек-
сте семантического web. Перспективным является использование мультимодельного подхода 
к построению и реализации сложных систем (например, информационных моделей киберфи-
зических систем), когда одна часть системы представляется с использованием одной модели 
(например, тех же РСП), другая часть – с использованием другой модели (например, функ-
циональных блоков стандарта IEC 61499), и эти модели интегрируются на базе онтологий. 

